

# Pengantar SAS/IML

*$\tau\rho$*

# Penggunaan SAS/IML

- Dimulai dengan pernyataan PROC IML dan diakhiri dengan pernyataan QUIT:

```
PROC IML;
```

```
:
```

```
:
```

```
QUIT;
```

# Contoh

## Bentuk Skalar

```
proc iml;  
x = 7;  
y = 21.44;  
z = .;  
a = 'Door';  
b = 'baaaaa';  
print x y z a b;
```

## Bentuk matriks

```
proc iml;  
reset print;  
a = {1 2, 3 4};  
b = {5 6, 7 8};  
c = {1 2 3, 4 5 6};  
d = {7, 8, 9};  
e = {10 11 12};  
kata = {'ambil' 'simpan' 'pergi'};
```

Operasi : \* perkalian matriks, # perkalian unsur yg sepadan, ` transpose,  
|| gabung kolom, // gabung baris

# Latihan 1

- Misalkan diketahui matriks

$$X = \begin{bmatrix} 1 & 2 & 0 \\ 0 & 5 & . \end{bmatrix}$$

- Dari matriks  $X$  buatlah matriks-matriks berikut ini:

$$- Y_1 = \begin{bmatrix} 0 & 0 & 1 \\ 1 & 0 & 0 \end{bmatrix}$$

$$- Y_2 = \begin{bmatrix} 1 & 0 & 1 \\ 1 & 0 & 1 \end{bmatrix}$$

$$- Y_3 = \begin{bmatrix} 1 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

# Jawaban 1

```
x = { 1 2 0, 0 5 . };  
y1 = (x = 0);  
y2 = (x <= 1);  
y3 = (x < 4 & x ^= 0);
```

# Latihan 2

- Diketahui SPL sebagai berikut:

$$3x_1 + 2x_2 - x_3 = -2$$

$$-3x_1 - x_2 + x_3 = 5$$

$$3x_1 + 2x_2 + x_3 = 2$$

- Tentukan solusinya dengan SAS/IML

# Jawaban 2

```
proc iml;  
  reset print;  
  A = {3 2 -1,  
       -3 -1 1,  
        3 2 1};  
  b = {-2, 5, 2};  
  x = inv(A) * b;
```

# Latihan 3

- Diketahui data regresi sebagai berikut:

Y	11	13	24	29	35	37
X	2	5	6	8	10	11

- Tentukan dugaan koefisien regresi dengan menggunakan SAS/IML

(hint :  $\underline{b} = (X'X)^{-1}(X'y)$ )



# Jawaban 3

```
proc iml;  
  reset print;  
  X = {2, 5, 6, 8, 10, 11};  
  y = {11, 13, 24, 29, 35, 37};  
  
  Xb = J(6, 1, 1) || X;  
  koef = inv(Xb` * Xb) * (Xb` * y);
```

# Latihan 4

- Diketahui matriks sebagai berikut

$$A = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

- Tentukan nilai rata-rata dari setiap baris dan kolom matriks A (dengan perkalian matriks)

# Jawaban 4

```
proc iml;  
  reset print;  
  A = {1 2 3,4 5 6,7 8 9};  
  s = {1, 1, 1};  
  
  rataanbaris = (1/3)*A*s;  
  rataankolom = (1/3)*s`*A;
```

# Latihan 5

- Diketahui matriks sebagai berikut

$$A = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

- Ubahlah A menjadi matriks Y, yaitu

$$Y = \begin{bmatrix} 3 & 6 & 9 \\ 2 & 5 & 8 \\ 1 & 4 & 7 \end{bmatrix}$$

# Jawaban 5

```
proc iml;  
reset print;  
A = {1 2 3,4 5 6,7 8 9};  
S = {0 0 1,0 1 0, 1 0 0};  
  
X = A*S;  
Y = X`;
```

# Latihan 6

- Buatlah tabel ringkasan yang terdiri dari total dan rata-rata terboboti pada matriks nilai berikut

Nama	UTS	UAS	Tugas
Budi	82	80	85
Joko	76	77	75
Jeki	78	90	86

Catatan: rata-rata terboboti :  $40\%UTS+40\%UAS+20\%Tugas$

# Jawaban 6

```
proc iml;  
reset print;  
nilai = {82 80 85,76 77 75,78 90 86};  
nama = {"Budi", "Joko", "Jeki"};  
ujian = {"UTS" "UAS" "Tugas"};  
  
print nilai[r=nama c=ujian];  
  
total = nilai*J(3,1,1);  
s = {0.4,0.4,0.2};  
ratab = nilai*s;  
  
print nilai[r=nama c=ujian] total ratab;
```

Terima kasih 😊